```
Multichar_Symbols + +CONJPART +EMPTY +NEG +INTERR +REL +OBJ +MASC +AUGM +ROOT +W2
+SIMPLE +PRET +ABS +CONJ +1P +2P +3P +SG +PL +PASS ^INFV @U.CONJ.NO@ @U.CONJ.YES@
^END_UNKNOWN @U.InfixPronClass.A@ @U.InfixPronClass.C@

!************************************************************* DEPENDENCY
*****************************************************************

! HOW TO READ THIS:
! First of all, '!' is reserved for comments and is ignored by the intepreter...
! This file type / program works on the basis of specifying lexicons and referring
to continuation lexicons within those lexicons
! e.g. from Dependency we go down to Independent or Dependent, from Dependent to
ConjunctParticle etc. (order in file irrelevant)
! we add morphological material on the way, e.g. ní .... +AUGM:ro ...... +OBJ+2P
+SG:t ....... +bris etc. .........+PRET+ABS+1P+PL:em


LEXICON Dependency
        Independent;
        Dependent;

! Dependency: binary inflection patterns: depending on whether there is a prefix
(dependent, e.g. sentence particles such as ní='not', in=
! question particle) or not (independent). Conjunct inflection stems from the fact
that there is a (syntactical) conjunct (binding) particle
! in front of it. However, some 'non-syntactical' prefixes produce forms which are
not called dependent in the literature (even though conjunct
! inflection invariably applies), e.g. augment 'ro', a prefixed perfective marker
(may be infixed after a syntactical / conjunct particle
! prefix as well). I therefore named the ultimate continuation lexicon that leads
to the verb stems 'VerbWithPrefix',i.e. a verb that is
! preceded by something.


LEXICON Independent
@U.CONJ.NO@             Simple;
@U.CONJ.YES@            Augment;


!  requirements for ENDINGS: CONJ.NO = absolute inflection, CONJ.YES = conjunct
inflection

LEXICON Dependent
        ConjunctParticle;

LEXICON VerbWithPrefix
@U.CONJ.YES@            Simple;


!************************************************************* PREFIXES
*****************************************************************

        !**************(incl. augm ro, which is only a prefix with augmented
simple verbs, otherwise infix, see below) ******************

LEXICON ConjunctParticle
ní@U.InfixPronClass.A@          Infix;
in@U.InfixPronClass.C@          Infix;

!  This rather cumbersome way of coding continuation lexicons is because this
rigid system of classes of pronouns after certain
! particles collapses in the later medieval period, and using flag diacritics
might be better than repeating the particles and their resp.
! continuation lexicons... i.e. more elegant when I have to expand on my system to
cover later periods... Even if I won't end up using many
! flag diacritics, I would still be interested in solving this matter...
```

```
!************************************************************* INFIXES
*********************************************************************


LEXICON Infix
        InfixPron; ! infixed pronoun (usually direct object)
        Augment;
        VerbWithPrefix; ! 'escape'

! various infix sub-type lexicons loop back to main Infix lexicon. This is to
facilitate e.g. the position of augments and infixed pronoun
! class A (>objects, e.g. -t) which interchange (each occurring max. once, see
filter in script) e.g. níro-t· + dependent (prefixed) simple
! verb or ní-t·ro + dependent (prefixed) simple verb. see EIV ch9-10

LEXICON Augment
@U.InfixPronClass.A@            Ro;

LEXICON Ro
+AUGM:ro        Infix;

LEXICON InfixPron
@U.InfixPronClass.A@            InfixPronClassA;
@U.InfixPronClass.C@            InfixPronClassC;

LEXICON InfixPronClassA ! finish full paradigm later
+OBJ+1P+SG:m            Infix;
+OBJ+2P+SG:t            Infix;
+OBJ+3P+SG+MASC:^INFVa  Infix; ! delete preceding o (ro) when coalescing with 'a',
or delete this vowel (a) after ní.

LEXICON InfixPronClassC ! finish full paradigm later
+OBJ+1P+SG:dom          Infix;  ! + lenition
+OBJ+2P+SG:dat          Infix;  ! + lenition
+OBJ+3P+SG+MASC:id      Infix;  ! + nasalisation
+OBJ+3P+SG+MASC:did     Infix;  ! + nasalisation
+OBJ+3P+SG+MASC:d       Infix;  ! + nasalisation
+OBJ+3P+SG+MASC:da      Infix;  ! + nasalisation


!************************************************************** STEMS
*********************************************************************

LEXICON Simple
        SimpleW2; ! Weak type 2 verbs, example root bris 'break'

LEXICON SimpleW2
+bris+ROOT+W2+SIMPLE:bris       W2Endings;


LEXICON W2Endings
        PrimEnd_sPret_Pal;

! one inflectional example: the preterite (narrative past) which can be augmented
with prefixed 'ro' to produce a perfect

!************************************************************* ENDINGS
*********************************************************************

LEXICON PrimEnd_sPret_Pal       ! s-preterite, with a PALATAL as opposed to a NON-
PAL stem consonant s.
@U.CONJ.NO@     PrimEnd_sPret_PalAbs;
@U.CONJ.YES@    PrimEnd_sPret_PalConj;
```

```
LEXICON PrimEnd_sPret_PalAbs
+PRET+ABS+1P+SG:siu             #;
+PRET+ABS+2P+SG:si              #;
+PRET+ABS+3P+SG:is              #;
+PRET+ABS+3P+SG+REL:es          #;
+PRET+ABS+1P+PL:simmi           #;
+PRET+ABS+1P+PL+REL:simme       #;
+PRET+ABS+2P+PL:^END_UNKNOWN    #;
+PRET+ABS+3P+PL:sit             #;
+PRET+ABS+3P+PL+REL:site        #;

LEXICON PrimEnd_sPret_PalConj
+PRET+CONJ+1P+SG:ius            #;
+PRET+CONJ+2P+SG:is             #;
+PRET+CONJ+3P+SG:0              #;
+PRET+CONJ+1P+PL:sem            #;
+PRET+CONJ+2P+PL:sid            #;
+PRET+CONJ+3P+PL:set            #;
```